


Funciones relacionales: RELATED y RELATEDTABLE

Dos funciones útiles que puede utilizar para navegar a través de las relaciones dentro de una fórmula DAX son RELATED y RELATEDTABLE.

Si observamos este modelo de datos:



Con respecto a las cadenas de relaciones, **todas las relaciones deben ser del mismo tipo, es decir, de uno a muchos o de muchos a uno**. Si la cadena enlaza dos tablas a través de una relación de uno a varios con una **tabla puente**, seguida de una relación de varios a uno con la segunda tabla, entonces ni RELATED ni RELATEDTABLE funcionan con propagación de filtro en una sola dirección. **Solo RELATEDTABLE puede funcionar usando propagación de filtro bidireccional**, como se explica más adelante. Por otro lado, una **relación de uno a uno** se comporta como una relación de uno a muchos y como una relación de muchos a uno al mismo tiempo. Por lo tanto, puede haber una relación uno a uno en una cadena de uno a muchos (o de muchos a uno) sin interrumpir la cadena.

Por ejemplo, en el modelo que elegimos como referencia, Cliente está relacionado con Ventas y Ventas está relacionado con Producto. Existe una relación de uno a varios entre el cliente y las ventas, y luego una relación de varios a uno entre las ventas y el producto. Por lo tanto, una cadena de relaciones vincula al Cliente con el Producto. Sin embargo, las dos relaciones no van en la misma dirección. **Este escenario se conoce como relación de varios a varios**. Un cliente está

relacionado con muchos productos comprados y, a su vez, un producto está relacionado con muchos clientes que compraron ese producto. Si se usa RELATEDTABLE a través de una relación de varios a varios, el resultado sería incorrecto.

```
Product[NumOfBuyingCustomers] =  
VAR CustomersOfCurrentProduct = RELATEDTABLE ( Customer )  
RETURN  
COUNTROWS ( CustomersOfCurrentProduct )
```

El resultado del código anterior no es el número de clientes que compraron ese producto. En cambio, el resultado es el número total de clientes:

Product Name	NumOfBuyingCustomers
A. Datum Advanced Digital Camera M300 Azure	18869
A. Datum Advanced Digital Camera M300 Black	18869
A. Datum Advanced Digital Camera M300 Green	18869
A. Datum Advanced Digital Camera M300 Grey	18869
A. Datum Advanced Digital Camera M300 Orange	18869
A. Datum Advanced Digital Camera M300 Pink	18869
A. Datum Advanced Digital Camera M300 Silver	18869
A. Datum All in One Digital Camera M200 Azure	18869
A. Datum All in One Digital Camera M200 Black	18869
A. Datum All in One Digital Camera M200 Green	18869

La tabla **Producto** no puede filtrar a la tabla **Customer**, mientras que, la tabla **Customer** si puede filtrar a la tabla **Producto**. Esto por las direcciones de filtro cruzado en el modelo de datos.

RELATEDTABLE no puede seguir la cadena de relaciones porque no van en la misma dirección. El contexto de fila del Producto no llega a los Clientes. Cabe señalar que, si probamos la fórmula en sentido contrario, es decir, si contamos el número de productos comprados para cada cliente, el resultado es correcto: un número diferente para cada fila que representa el número de productos comprados por el cliente. La razón de este comportamiento no es la propagación de un contexto de fila sino, más bien, la transición de contexto generada por RELATEDTABLE.


Con el contexto de fila, usamos RELATED y RELATEDTABLE para propagar el contexto de fila a través de relaciones. Por otro lado, con el contexto de filtro, no se necesitan funciones para propagar el filtro. El contexto de filtro filtra el modelo, no una tabla. Como tal, una vez que se aplica un contexto de filtro, todo el modelo está sujeto al filtro de acuerdo con las relaciones.

1. RELATED

= RELATED(columna)

RELATED sigue una relación de **uno a muchos** y devuelve el valor relacionado de la otra tabla.

No habrá problemas en traer dos campos a la Tabla Pedido, un campo de la Tabla SKUProductos y otro de CategoriaProductos. Dado que la Información fluye en relación “uno a muchos” no habrá problemas.



[RELATED]		fx=RELATED(CategoriaProductos[Categoría de Producto])						
	Número de Pedido	Tipo de Compra	S..	Categoría de Descuento	Precio de Venta	Unidades	RELATED	
1	98069378	Normal	C036	Day Off Full	71950	10	Arte	
2	98069388	Devolucion	C018	Day Off Full	56260	10	Música	
3	98069400	Devolucion	C004	Day Off Full	9458	2	Cine	
4	98069404	Devolucion	C015	Day Off Full	28836	9	Videojuegos	
5	98069406	Devolucion	C036	Day Off Full	71950	10	Arte	
6	98069410	Normal	C020	Day Off Full	32805	5	Música	
7	98069411	Devolucion	C005	Day Off Full	37610	10	Cine	
8	98069412	Devolucion	C035	Day Off Full	37584	9	Cine	
9	98069415	Devolucion	C014	Day Off Full	16550	2	Cine	
10	98069417	Devolucion	C030	Day Off Full	7342	2	Arte	
11	98069419	Devolucion	C018	Day Off Full	45008	8	Música	
12	98069421	Normal	C019	Day Off Full	28452	4	Cine	
3	98069400	Devolucion	C004	Day Off Full	9458	2	Producto4	Cine
4	98069404	Devolucion	C015	Day Off Full	28836	9	Producto15	Videojuegos
5	98069406	Devolucion	C036	Day Off Full	71950	10	Producto36	Arte
6	98069410	Normal	C020	Day Off Full	32805	5	Producto20	Música
7	98069411	Devolucion	C005	Day Off Full	37610	10	Producto5	Cine
8	98069412	Devolucion	C035	Day Off Full	37584	9	Producto35	Cine

Ejemplo 1: IF / RELATED (Columna calculada)

```
Sales[AdjustedCost] =  
IF (  
    RELATED ( 'Product Category'[Category] ) = "Cell Phone",  
    Sales[Unit Cost] * 0.95,  
    Sales[Unit Cost]  
)
```

Ejemplo 2: Medida

Debe existir:

1. Contexto de fila
2. Tablas deben estar relacionadas
3. El llamado debe ser desde una Tabla de hechos hacia una Tabla dimensión

```
1 UtldRLista =  
2 VAR PedidosNormal =  
3     FILTER(  
4         Pedidos ;  
5         Pedidos[Tipo de Compra] = "Normal"  
6         && Pedidos[SKU] IN 'Unidades_Almacén'  
7     )  
8 VAR Utilidad =  
9     SUMX(  
10         PedidosNormal ;  
11         VAR PctDescuento =  
12             RELATED( Descuentos[Descuento] )  
13         VAR CostoTotal =  
14             Pedidos[Costo de Envio]  
15             + Pedidos[Costo del Producto]  
16             + Pedidos[Costo Empaque]  
17         VAR CalculoDeUtilidad =  
18             ( Pedidos[Ingresos]* ( 1- PctDescuento ) ) - CostoTotal  
19         RETURN  
20             CalculoDeUtilidad  
21     )  
22 RETURN  
23     Utilidad
```

Ejemplo 3:

```

1 IngresosReales =
2     SUMX(
3         FILTER(
4             Pedidos,
5             Pedidos[Tipo de Compra] = "Normal"
6         ),
7         Pedidos[Ingresos] * (1 - RELATED(Descuentos[Descuento]))
8     )

```

Nombre del Producto	Ingresos Tot	IngresosReales
50 Must-Have Beethoven Masterpieces	19.387	14.822
Batman Begins	45.486	34.071
Batman Begins + The Dark Knight + The Dark Knight Rises	1.885.235	1.412.600
Beethoven: Complete Symphonies	28.658	21.928
Calculus - Ron Larson	78.914	60.204
Calculus and Analytic Geometry - George B	53.646	39.970
Calculus For Dummies - Mark	106.404	78.953
Calculus, 7th Edition - James	171.819	131.426
Cocktail Party Jazz	18.755	14.153
Green Hill Jazz Sampler	17.284	13.373
Inception	24.044	18.097
Interstellar	44.571	33.524
Jazz Collection	13.765	11.092
Men of Steel	30.156	23.302
Physics for Scientists and Engineers - Serway	103.313	77.520
Sabaton: Heroes + Sabaton: Coar Arms + Sabaton: Caralus Rex	313.914	240.152
Sabaton: The Art of War + Sabaton: Primo Victoria	109.568	83.377
Sears & Zemansky's University Physics, Vol. 1	143.427	108.005
Sears & Zemansky's University Physics, Vol. 2	55.013	41.543
The Dark Knight	26.486	19.833
The Dark Knight Rises	20.275	15.283
Total	3.320.219	2.500.229

2. RELATEDTABLE

= RELATEDTABLE(Tabla)

Sigue una relación en ambos sentidos (“*uno a muchos*” y “*muchos a uno*”) y **devuelve una tabla que contiene todas las filas que están relacionadas con la fila actual de la Tabla especificada**. Esto es muy útil cuando necesitamos encontrar todas las transacciones asociadas con una fila determinada de una tabla relacionada.

Tener en cuenta que esta función devuelve una Tabla y no un valor escalar. Esto significa que esta función no puede utilizarse por sí misma para definir una columna calculada. En cambio, esta función puede utilizarse para proporcionar un resultado intermedio que a su vez es un argumento para otra función, como una función de agregación.

Al utilizarse SUMX sumará todos los Ingresos que se trajeron desde la tabla de hechos “Pedidos” con RELATEDTABLE a la Tabla de dimensión “Productos”:

[RELATEDTABL...		fx =SUMX(RELATEDTABLE(Pedidos), Pedidos[Ingresos])			
	S...	Producto	Precio de Venta	RELATEDTABLE	Agregar colum
1	C001	Producto1	2743	414193	
2	C002	Producto2	6405	858270	
3	C003	Producto3	7755	1155495	
4	C004	Producto4	4729	534377	
5	C005	Producto5	3761	699546	
6	C006	Producto6	3530	444780	
7	C007	Producto7	7308	971964	
8	C008	Producto8	7926	903564	
9	C009	Producto9	8994	1349100	
10	C010	Producto10	7846	917982	
11	C011	Producto11	6362	718906	
12	C012	Producto12	2777	386003	
13	C013	Producto13	7996	751624	
14	C014	Producto14	8275	1787400	
15	C015	Producto15	3204	611964	
16	C016	Producto16	4912	741712	
17	C017	Producto17	4919	550928	

2.1. COUNTROWS / RELATEDTABLE (Columna calculada)

Si queremos saber cuántos productos hay en cada categoría, podemos crear una columna en *Product Category* con esta fórmula:

```
'Product Category'[NumOfProducts] = COUNTROWS ( RELATEDTABLE ( Product ) )
```

Category	NumOfProducts
Audio	115
Cameras and camcorders	372
Cell phones	285
Computers	606
Games and Toys	166
Home Appliances	661
Music, Movies and Audio Books	90
TV and Video	222

En una columna calculada o dentro de una iteración, también se puede usar la función RELATEDTABLE para recuperar todas las filas de una tabla relacionada. Por ejemplo, la siguiente columna calculada en la tabla *Product* calcula el monto de ventas del producto correspondiente:

```
'Product'[Product Sales Amount] =  
SUMX (  
    RELATEDTABLE ( Sales ),  
    Sales[Quantity] * Sales[Unit Price]  
)
```

Category	CategorySales
Audio	\$384,518.16
Cameras and camcorders	\$7,192,581.95
Cell phones	\$1,604,610.26
Computers	\$6,741,548.73
Games and Toys	\$360,652.81
Home Appliances	\$9,600,457.04
Music, Movies and Audio Books	\$314,206.74
TV and Video	\$4,392,768.29

Nota

La única excepción a la regla general de RELATED y RELATEDTABLE es para las relaciones uno a uno. Si dos tablas comparten una relación uno a uno, tanto RELATED como RELATEDTABLE funcionan en ambas tablas y dan como resultado un valor de columna o una tabla con una sola fila, según la función utilizada.

2.2. SUMX / RELATEDTABLE (Columna calculada)

Por ejemplo, la siguiente columna calculada en la tabla *Product* calcula el monto de ventas del producto correspondiente:

```
'Product'[Product Sales Amount] =  
SUMX (  
    RELATEDTABLE ( Sales ),  
    Sales[Quantity] * Sales[Unit Price]  
)
```

2.3. SUMX / FILTER / RELATEDTABLE (Columna calculada)

Las funciones de tabla también se pueden anidar. Por ejemplo, la siguiente columna calculada en la tabla *Product* calcula el monto de las ventas del producto considerando solo las ventas con una cantidad mayor que uno:

```
'Product'[Product Sales Amount Multiple Items] =  
SUMX (  
    FILTER (  
        RELATEDTABLE ( Sales ),  
        Sales[Quantity] > 1  
    ),  
    Sales[Quantity] * Sales[Unit Price]  
)
```

En el código de muestra, RELATEDTABLE está anidado dentro de FILTER. Como regla general, cuando hay llamadas anidadas, DAX evalúa primero la función más interna y luego evalúa las demás hasta la función más externa.